
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 6.2 which is part of the IMS NR5GC test suite.

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.
Contact:
Kalahasti, Narendra

Narendra.Kalahasti@anritsu.com
1.1 Verification Test Summary 
Test Case: 
6.2
ATS Version:
iwd-TTCN3-B2019-06_D20wk38
System Simulator used:
Anritsu Protocol Conformance Test System ME7834NR
UE used:
Qualcomm SM8350
Verification Status:
PASS
2 Corrections required
2.1 Change 1

	Test case name
	6.2

	Reason for change
	Based on [TS 24.229 cl. 5.1.1.2.1]
For each 4xx, 5xx or 6xx response received without a Retry-After header field to the REGISTER request, the UE shall:

a) mark the currently used P-CSCF address as unavailable for the last duration of the retry delay time computed by the algorithm defined in subclause 4.5 of RFC 5626 [92] plus 5 minutes; and

When a UE receives 4xx,5xx or a 6xx response without a retry-after header, and there is no other P-CSCF available, UE behaviour is not defined in TS 24.229 cl 5.1.1.2, so it is up to UE implementation to consider this as a case for “IMS voice not available “ as per TS 24.501 Cl 4.3 and 4.9.2 and the UE may disable N1 mode capability. Thus one P-CSCF config may fail a conformant UE. Due to the problem described above and to keep the testability of the conforamance requirement related to TP1, a second P-CSCF is introduced in the test configuration.Test behavior is modified accordingly.
The REGISTER messages at step 4, 6 and 8 will be sent over a new dialog, being handled on a separate PDN by separate IMS component. We propose to use PDN_3 (IMS3) for this purpose since traditionally the PDN_2 is used for multiPDN scenarios in LTE/EN-DC test cases so to differentiate the purpose other than PDN_2 was used, i.e. PDN_3
R5-205581 raised at RAN5meeting#89-e is associated with this CR.


	Summary of change
	Second IMS component associated with PDN_3 has been established for this test where all steps after 4th one will be handled.

	TTCN module
	IMS_Testsuite_NR5GC.ttcn

	MCC160 Comment
	


Before change:
  testcase TC_6_2_NR5GC() runs on MTC_IMS system SYSTEM_IMS {

    // @purpose

    //   Initial Registration Failures / 5GS

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(600);

    v_IMS1 := IMS_PTC.create alive;

    f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(system, v_IMS1, v_IMS2);

    v_IMS1.start(f_TC_6_2_NR5GC_IMS1());

    t_GuardTimer.start;

    f_MTC_MainLoop_IMS(t_GuardTimer);

  }
After change:
        testcase TC_6_2_NR5GC() runs on MTC_IMS system SYSTEM_IMS {

    // @purpose

    //   Initial Registration Failures / 5GS

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    var IMS_PTC          v_IMS3       := null;

    timer t_GuardTimer := int2float(600); 
    v_IMS1 := IMS_PTC.create alive;

    v_IMS3 := IMS_PTC.create alive;

    f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(system, v_IMS1, v_IMS2, -, -, v_IMS3);

    v_IMS1.start(f_TC_6_2_NR5GC_IMS1());

    v_IMS3.start(f_TC_6_2_NR5GC_IMS3());

    t_GuardTimer.start;

    f_MTC_MainLoop_IMS(t_GuardTimer);

  }
2.2 Change 2

	Constant name
	tsc_NoOfIMSPDNs
tsc_Index_IMS3

	Reason for change
	Enabling additional IMS PDN requires to increase the constant tsc_NoOfIMSPDNs by one. Selecting the PDN_3 for the additional IMS PDN requires to create new constant for the IMS index.

	Summary of change
	tsc_NoOfIMSPDNs has been increased by 1 and new tsc_Index_IMS3 constant has been created.

	TTCN module
	Common\CommonDefs.ttcn

	MCC160 Comment
	


Before change:
    const integer tsc_NoOfIMSPDNs := 2;
After change:

      const integer tsc_NoOfIMSPDNs := 3;
      const integer tsc_Index_IMS3 := 2; //added
2.3 Change 3

	Function name
	f_GetTestcaseAttrib_AdditionalPCSCF

	Reason for change
	In order to accommodate that certain changes are applied only in case a testcase needs an additional P-CSCF, the new testcase attribute needs to be created.

	Summary of change
	New testcase attribute “additionalPCSCF” has been defined

	TTCN module
	Common\TestcaseProperties.ttcn

	MCC160 Comment
	


After change:

  function f_GetTestcaseAttrib_AdditionalPCSCF(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_6_2_NR5GC") { return true; }

    }

    return false;

  }
2.4 Change 4

	Function name
	f_NR5GC_PDUSessionEstablishment_InnerLoop

	Reason for change
	In case a testcase requires additional P-CSCF, also the IP handling for PDN_3 needs to be started.

	Summary of change
	IP handling of PDN_3 is started during PDU establishment procedure in case of additional P-CSCF testcase.

	TTCN module
	Common\NR5GC\NR5GC_NASSteps.ttcn

	MCC160 Comment
	


Before change:
    function f_NR5GC_PDUSessionEstablishment_InnerLoop(NR_CellId_Type p_CellId,

                                                     NR_SRB_COMMON_IND p_ReceivedAsp,

                                                     SRBs_AlreadyConfigured p_SRB := SRB1_2,

                                                     template (value) SDAP_Config p_SDAP_Config := cs_SDAP_Config,

                                                     template S_NSSAI_Type p_ExpectedS_NSSAI := *,

                                                     template DNN p_ExpectedDNN := *,

                                                     template (value) SSC_Mode p_SSC_Mode := cs_SSC_Mode(-, '001'B),

                                                     template (omit) DNN p_DNN := omit,

                                                     template (omit) QoS_Rules p_QoS_Rules := omit,

                                                     template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),

                                                     template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@

                                                     template (omit) QoSFlowDescr p_QoSFlowDescr := omit,

                                                     template (present) O1_Type p_PDU_SessionId := ?,

                                                     template ProcedureTransactionIdentifier p_PTI := omit,

                                                     template (omit) PDU_SessionType p_PDU_SessionType := omit,

                                                     template SSC_Mode p_ExpectedSSC_Mode := *, // @sic R5s190290 sic@

                                                     template NG_UE_SM_Cap p_UECap := *,

                                                     template MaxNumPacketFilters p_MaxNumPacketFilters := *,

                                                     template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := ?,

                                                     template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,

                                                     template SM_PDU_DN_RequestContainer p_PDUReq := *,

                                                     template ExtdProtocolConfigOptions p_ExtdPCO := *,

                                                     template NG_PDU_SessionId p_OldPDU_SessionId := *,

                                                     template NG_Request_Type p_RequestType := ?,

                                                     template AdditionalInformation p_AdditionalInfo := *,

                                                     template (omit) GMM_GSM_Cause p_Cause := omit,

                                                     template (omit) GPRS_Timer p_RQTimer := omit,

                                                     template (omit) EAP_Message p_EAP := omit,

                                                     template (omit) GPRS_Timer3 p_BackOff := omit,

                                                     boolean p_PDUSessionAuthenticationRequired := false) runs on NR5GC_PTC // @sic R5-198184 sic@

  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@

    // Moved UT trigger up to f_NR5GC_PDUSessionEstablishment @sic R5s200751 sic@

    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo(); // @sic R5-195328 sic@

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template (value) CellGroupConfig v_CellGroupConfig;

    var template (value) NR_RadioBearerList_Type v_SSConfig;

    var template (value) QosFlow_Identification_Type v_QosFlowId;

    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;

    var SdapConfig_Type v_SdapConfig;

    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (p_CellId);

    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);

    var SDAP_Config v_SDAP_Config := valueof(p_SDAP_Config); //@sic R5-190720 sic@

    var template (omit) PDN_Index_Type v_PDN_Index := omit;

    var NG_NAS_SecurityParams_Type v_SecurityParams; // @sic R5-198184 sic@

    if (not f_Check_NG_PDUSessionEstablishmentReq(v_GSM_MobilityInfo,

                                                  p_ReceivedAsp.Signalling.Nas[0].Pdu,

                                                  p_PDU_SessionId,

                                                  p_PTI,

                                                  p_IntegrityProtMaxDataRate,

                                                  p_PDU_SessionType,

                                                  p_ExpectedSSC_Mode,

                                                  p_UECap,

                                                  p_MaxNumPacketFilters,

                                                  p_AlwaysOnPDUSessionReq,

                                                  p_PDUReq,

                                                  p_ExtdPCO,

                                                  p_OldPDU_SessionId,

                                                  p_RequestType,

                                                  p_ExpectedS_NSSAI,

                                                  p_ExpectedDNN,

                                                  p_AdditionalInfo)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Establishment Message Failed");

    }

    f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,

                               v_CellGroupConfig,

                               v_SSConfig,

                               v_SDAP_ConfigurationList,

                               p_CellId,

                               {v_GSM_MobilityInfo},

                               p_SRB,

                               v_SDAP_Config);

    // Only one PDU session input, so only 1st element relevant in output

    v_SdapConfig := valueof(v_SDAP_ConfigurationList[0].Config.SdapConfig);

    //It is assumed that by default only 1 QoS Flow is specified

    v_QosFlowId := cs_IP_QosFlowId(v_SdapConfig.Pdu_SessionId, v_SdapConfig.MappedQoS_Flows[0]);

    // If this is a default DRB, enable IP handling for this DRB

    if (v_SDAP_Config.defaultDRB) {

      v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo);

      f_IP_Handling_StartPDN(IP, valueof(v_PDN_Index), cs_DrbInfo_NR(p_CellId, v_QosFlowId)); // @sic R5s201094 sic@

    }

    if(p_PDUSessionAuthenticationRequired){ // @sic R5-198184 sic@

    //Perform PDU Session Authentication if required

    v_SecurityParams := f_NR5GC_Security_Get();

    f_PDUSessionAuthentication( p_CellId, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN);

    }

    // @sic R5-204400 sic@

    v_GSM_MobilityInfo.EPS_Bearer := valueof(f_GetNextEPSBearerId (v_GSM_MobilityInfo, v_GMM_MobilityInfo));

    v_GSM_MobilityInfo.PdnIndex := valueof(v_PDN_Index);

    // Send Accept

    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionEstablishmentAccept (v_GSM_MobilityInfo,

                                                                      v_GMM_MobilityInfo,

                                                                      v_PLMN,

                                                                      p_SSC_Mode,

                                                                      p_Session_AMBR,

                                                                      p_PDU_SessionType,

                                                                      p_Cause,

                                                                      p_RQTimer,

                                                                      p_S_NSSAI,

                                                                      -,

                                                                      p_EAP,

                                                                      p_QoS_Rules,

                                                                      p_QoSFlowDescr,

                                                                      omit,

                                                                      p_DNN,

                                                                      p_BackOff);

    // Add this PDU Session to stored list

    f_NR5GC_MobileInfo_AddPDUSession (v_GSM_MobilityInfo);

    // Send RRC Reconfiguration message

    f_NR_RRCReconfigNewDRBWithSSConfig(p_CellId,

                                       v_RadioBearerConfig,

                                       v_SSConfig,

                                       v_CellGroupConfig,

                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) }); // @sic R5s190629 sic@

  }

  /*

   * @desc      If the UE supports EUTRA, get the next EPS Bearer Id, if not return omit

   * @param     p_GSM_MobilityInfo

   * @param     p_GMM_MobilityInfo

   * @return    template (omit) integer

   * @status    APPROVED (NR5GC, NR5GC_IRAT)

   */

  function f_GetNextEPSBearerId (GSM_MobilityInfo_Type p_GSM_MobilityInfo,

                                 GMM_MobilityInfo_Type p_GMM_MobilityInfo) runs on NR5GC_PTC return template (omit) integer

  {

    var template (omit) integer v_BearerId := omit;

    var IntegerList_Type v_ExistingIds := f_NR5GC_MobileInfo_GetEPS_BearerIdList();

    var integer v_NoOfIds := lengthof(v_ExistingIds);

    var integer i;

    if (isvalue(p_GMM_MobilityInfo.GMMCap)) {

      if (p_GMM_MobilityInfo.GMMCap.s1Cap == '1'B) {

        if (p_GSM_MobilityInfo.IMS) {

          v_BearerId := 5;

          for (i:= 0; i < v_NoOfIds; i :=i+1) {

            if (match (v_ExistingIds[i], 5)) { // IMS Bearer id = 5

              v_BearerId := v_NoOfIds + (6-1); // allocate the next one, from 6 onwards

              break;

            }

          }

        } else { // Not IMS

          v_BearerId := v_NoOfIds + 6;

        }

        // Add the new id to the end of the list and store it

        v_ExistingIds[v_NoOfIds] := valueof(v_BearerId);

        f_NR5GC_MobileInfo_SetEPS_BearerIdList(v_ExistingIds);

      }

    }

    return v_BearerId;

  }
After change:

  import from TestcaseProperties all; //f_GetTestcaseAttrib_AdditionalPCSCF
   function f_NR5GC_PDUSessionEstablishment_InnerLoop(NR_CellId_Type p_CellId,

                                                     NR_SRB_COMMON_IND p_ReceivedAsp,

                                                     SRBs_AlreadyConfigured p_SRB := SRB1_2,

                                                     template (value) SDAP_Config p_SDAP_Config := cs_SDAP_Config,

                                                     template S_NSSAI_Type p_ExpectedS_NSSAI := *,

                                                     template DNN p_ExpectedDNN := *,

                                                     template (value) SSC_Mode p_SSC_Mode := cs_SSC_Mode(-, '001'B),

                                                     template (omit) DNN p_DNN := omit,

                                                     template (omit) QoS_Rules p_QoS_Rules := omit,

                                                     template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),

                                                     template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@

                                                     template (omit) QoSFlowDescr p_QoSFlowDescr := omit,

                                                     template (present) O1_Type p_PDU_SessionId := ?,

                                                     template ProcedureTransactionIdentifier p_PTI := omit,

                                                     template (omit) PDU_SessionType p_PDU_SessionType := omit,

                                                     template SSC_Mode p_ExpectedSSC_Mode := *, // @sic R5s190290 sic@

                                                     template NG_UE_SM_Cap p_UECap := *,

                                                     template MaxNumPacketFilters p_MaxNumPacketFilters := *,

                                                     template IntegrityProtMaxDataRate p_IntegrityProtMaxDataRate := ?,

                                                     template AlwaysOnPDUSessionReq p_AlwaysOnPDUSessionReq := *,

                                                     template SM_PDU_DN_RequestContainer p_PDUReq := *,

                                                     template ExtdProtocolConfigOptions p_ExtdPCO := *,

                                                     template NG_PDU_SessionId p_OldPDU_SessionId := *,

                                                     template NG_Request_Type p_RequestType := ?,

                                                     template AdditionalInformation p_AdditionalInfo := *,

                                                     template (omit) GMM_GSM_Cause p_Cause := omit,

                                                     template (omit) GPRS_Timer p_RQTimer := omit,

                                                     template (omit) EAP_Message p_EAP := omit,

                                                     template (omit) GPRS_Timer3 p_BackOff := omit,

                                                     boolean p_PDUSessionAuthenticationRequired := false) runs on NR5GC_PTC // @sic R5-198184 sic@

  { // Split function so list of DRBs can be built elsewhere @sic R5-199103 sic@

    // Moved UT trigger up to f_NR5GC_PDUSessionEstablishment @sic R5s200751 sic@

    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo(); // @sic R5-195328 sic@

    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template (value) CellGroupConfig v_CellGroupConfig;

    var template (value) NR_RadioBearerList_Type v_SSConfig;

    var template (value) QosFlow_Identification_Type v_QosFlowId;

    var template (value) SDAP_ConfigurationList_Type v_SDAP_ConfigurationList;

    var SdapConfig_Type v_SdapConfig;

    var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (p_CellId);

    var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);

    var SDAP_Config v_SDAP_Config := valueof(p_SDAP_Config); //@sic R5-190720 sic@

    var template (omit) PDN_Index_Type v_PDN_Index := omit;

    var NG_NAS_SecurityParams_Type v_SecurityParams; // @sic R5-198184 sic@

    if (not f_Check_NG_PDUSessionEstablishmentReq(v_GSM_MobilityInfo,

                                                  p_ReceivedAsp.Signalling.Nas[0].Pdu,

                                                  p_PDU_SessionId,

                                                  p_PTI,

                                                  p_IntegrityProtMaxDataRate,

                                                  p_PDU_SessionType,

                                                  p_ExpectedSSC_Mode,

                                                  p_UECap,

                                                  p_MaxNumPacketFilters,

                                                  p_AlwaysOnPDUSessionReq,

                                                  p_PDUReq,

                                                  p_ExtdPCO,

                                                  p_OldPDU_SessionId,

                                                  p_RequestType,

                                                  p_ExpectedS_NSSAI,

                                                  p_ExpectedDNN,

                                                  p_AdditionalInfo)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Establishment Message Failed");

    }

    f_BuildRRCReconfigParamsNewDRBs (v_RadioBearerConfig,

                               v_CellGroupConfig,

                               v_SSConfig,

                               v_SDAP_ConfigurationList,

                               p_CellId,

                               {v_GSM_MobilityInfo},

                               p_SRB,

                               v_SDAP_Config);

    // Only one PDU session input, so only 1st element relevant in output

    v_SdapConfig := valueof(v_SDAP_ConfigurationList[0].Config.SdapConfig);

    //It is assumed that by default only 1 QoS Flow is specified

    v_QosFlowId := cs_IP_QosFlowId(v_SdapConfig.Pdu_SessionId, v_SdapConfig.MappedQoS_Flows[0]);

    // If this is a default DRB, enable IP handling for this DRB

    if (v_SDAP_Config.defaultDRB) {

      v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo);

      f_IP_Handling_StartPDN(IP, valueof(v_PDN_Index), cs_DrbInfo_NR(p_CellId, v_QosFlowId)); // @sic R5s201094 sic@

      if (f_GetTestcaseAttrib_AdditionalPCSCF (testcasename())) {

         f_IP_Handling_StartPDN(IP, valueof(PDN_3), cs_DrbInfo_NR(p_CellId, v_QosFlowId));

      } 

    }

    if(p_PDUSessionAuthenticationRequired){ // @sic R5-198184 sic@

    //Perform PDU Session Authentication if required

    v_SecurityParams := f_NR5GC_Security_Get();

    f_PDUSessionAuthentication( p_CellId, v_SecurityParams, v_GSM_MobilityInfo, v_PLMN);

    }

    // @sic R5-204400 sic@

    v_GSM_MobilityInfo.EPS_Bearer := valueof(f_GetNextEPSBearerId (v_GSM_MobilityInfo, v_GMM_MobilityInfo));

    v_GSM_MobilityInfo.PdnIndex := valueof(v_PDN_Index);

    // Send Accept

    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionEstablishmentAccept (v_GSM_MobilityInfo,

                                                                      v_GMM_MobilityInfo,

                                                                      v_PLMN,

                                                                      p_SSC_Mode,

                                                                      p_Session_AMBR,

                                                                      p_PDU_SessionType,

                                                                      p_Cause,

                                                                      p_RQTimer,

                                                                      p_S_NSSAI,

                                                                      -,

                                                                      p_EAP,

                                                                      p_QoS_Rules,

                                                                      p_QoSFlowDescr,

                                                                      omit,

                                                                      p_DNN,

                                                                      p_BackOff);

    // Add this PDU Session to stored list

    f_NR5GC_MobileInfo_AddPDUSession (v_GSM_MobilityInfo);

    // Send RRC Reconfiguration message

    f_NR_RRCReconfigNewDRBWithSSConfig(p_CellId,

                                       v_RadioBearerConfig,

                                       v_SSConfig,

                                       v_CellGroupConfig,

                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend) }); // @sic R5s190629 sic@

  }
2.5 Change 5

	altstep name
	a_NR5GC_StandardDefault

	Reason for change
	For the additional P-CSCF testcases there is a need to handle the IMS queries received on PDN_3 in the default behaviour for NR5GC component.

	Summary of change
	Added handling of IMS queries on PDN_3 for additional P-CSCF testcases.

	TTCN module
	NR5GC\NR_Component_NR5GC.ttcn

	MCC160 Comment
	


Before change:
     altstep a_NR5GC_StandardDefault() runs on NR5GC_PTC

  {

    [] NASCTRL.receive(cr_NG_NAS_CTRL_CNF_Any) {

      /* unexpected CNF shall always result in INCONC */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected receive event at port NASCTRL");

    }

    [] a_NR_IMS_QueryHandler(IMS[tsc_Index_IMS1]) {

      repeat;

    }

    [] a_NR_IMS_QueryHandler(IMS[tsc_Index_IMS2]) {

      repeat;

    }

    [] a_NR_BASE_StandardDefault() {

    }

  }
After change:

      import from TestcaseProperties all; //f_GetTestcaseAttrib_AdditionalPCSCF 

  altstep a_NR5GC_StandardDefault() runs on NR5GC_PTC

  {

    [] NASCTRL.receive(cr_NG_NAS_CTRL_CNF_Any) {

      /* unexpected CNF shall always result in INCONC */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected receive event at port NASCTRL");

    }

    [] a_NR_IMS_QueryHandler(IMS[tsc_Index_IMS1]) {

      repeat;

    }

    [] a_NR_IMS_QueryHandler(IMS[tsc_Index_IMS2]) {

      repeat;

    }

    [f_GetTestcaseAttrib_AdditionalPCSCF(testcasename())] a_NR_IMS_QueryHandler(IMS[tsc_Index_IMS3]) {

      repeat;

    }

    [] a_NR_BASE_StandardDefault() {

    }

  }
2.6 Change 6

	Function name
	f_Get_NG_PDUSessionEstablishmentAccept

	Reason for change
	In case of additional P-CSCF testcase the P-CSCF address defined for PDN_3 needs to be added into extended Protocol Configuration Options.
#Draft proseCR(TS38.508-1) would be raised at RAN5meeting#89-e

	Summary of change
	In case of additional P-CSCF testcase the P-CSCF address defined for PDN_3 is added into extended Protocol Configuration Options.

	TTCN module
	NG_NAS\NG_NASTemplateFunctions.ttcn

	MCC160 Comment
	


Before change:
       function f_Get_NG_PDUSessionEstablishmentAccept(GSM_MobilityInfo_Type p_GSM_MobilityInfo,

                                                  GMM_MobilityInfo_Type p_GMM_MobilityInfo,

                                                  NAS_PlmnId p_PLMN,

                                                  template (value) SSC_Mode p_SSC_Mode := cs_SSC_Mode(-, '001'B),

                                                  template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),

                                                  template (omit) PDU_SessionType p_PDU_SessionType := omit,

                                                  template (omit) GMM_GSM_Cause p_Cause := omit,

                                                  template (omit) GPRS_Timer p_RQTimer := omit,

                                                  template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@

                                                  template (omit) AlwaysOnPDUSessionInd p_AlwaysOnPDUSessionInd := omit,

                                                  template (omit) EAP_Message p_EAP := omit,

                                                  template (omit) QoS_Rules p_QoS_Rules := omit,

                                                  template (omit) QoSFlowDescr p_QoSFlowDescr := omit,

                                                  template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,

                                                  template (omit) DNN p_DNN := omit,

                                                  template (omit) GPRS_Timer3 p_BackOff := omit) return template (value) NG_NAS_DL_Pdu_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_GSMMsg;

    var template (value) PDU_SessionType v_PDU_SessionType;

    var B3_Type v_PDN_TypeToBeUsed;

    var PDN_AddressInfo_Type v_PDN_Address;

    var template (omit) PDU_Address v_PDU_Address := omit;

    var template (omit) AlwaysOnPDUSessionInd v_AlwaysOn := omit;

    var template (value) QoS_Rules v_QoS_Rules := f_BuildDefaultQoSRules ("1"); // @sic R5-202585 sic@

    var template (value) QoSFlowDescr v_QoSFlowDescr;

    var template (value) DNN v_DNN := cs_DNN(-, px_AccessPointName);

    var octetstring v_DNNValue;

    var template (omit) ExtdProtocolConfigOptions v_ExtdPCO;

    var template (omit) MappedEPSBearerContexts v_MappedEPSBearerContexts := omit;

    var template (value) MappedEPSContext v_MappedEPSContext;

    var bitstring v_EPSParam; // If used

    var O1_Type v_EPSBearerId; // If used

    var boolean v_InterworkWithEPS := false;

    var template (value) QoS_Flow v_QoS_Flow;

    if (isvalue(p_PDU_SessionType)) {

      v_PDU_SessionType := valueof(p_PDU_SessionType);

    } else {

      v_PDU_SessionType := cs_PDU_SessionType (p_GSM_MobilityInfo.SessionType);

    }

    if (isvalue(p_DNN)) { // test case is overwriting the DNN value

      v_DNN := valueof(p_DNN);

    } else if (pc_IMS_DNN_default or isvalue(p_GSM_MobilityInfo.DNN)) {   // if included with the Req

                                                      // @sic R5-204400 sic@ or not included but PICS = Default is IMS

      if (pc_IMS_DNN_default or p_GSM_MobilityInfo.IMS) { // overwrite value with complete response

        v_DNNValue := f_NG_GetDNNForIMS(p_GSM_MobilityInfo.DNN.dnnValue, p_PLMN);      //Anritsu TTCN CR R5s201327

      } else { // Not IMS, but included

        v_DNNValue := p_GSM_MobilityInfo.DNN.dnnValue;

      }

      v_DNN := cs_DNN(-, v_DNNValue);

    } // else use Default DNN already defined above if not included and not IMS

    if (isvalue(p_GMM_MobilityInfo.GMMCap)) { // @sic R5s190464 R5s191172 sic@

      if (p_GMM_MobilityInfo.GMMCap.s1Cap == '1'B) { // @sic R5-195328 sic@

        v_InterworkWithEPS := true;

        if (isvalue(p_GSM_MobilityInfo.EPS_Bearer)) { // @sic R5-204400 sic@

          v_EPSBearerId := hex2oct(int2hex(valueof(p_GSM_MobilityInfo.EPS_Bearer), 1) & '0'H);

        } else {

          FatalError(__FILE__, __LINE__, "EPS Supported but no bearer id calculated");

        }

        if (p_GSM_MobilityInfo.IMS) {

          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_2));

          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter(bit2oct(v_EPSParam)));

        } else {

          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_1));

          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter(bit2oct(v_EPSParam)));

        }

        v_MappedEPSBearerContexts := cs_OneMappedEPSBearerContextList (v_MappedEPSContext);

      }

    }

    if (isvalue(p_QoS_Rules)) {

      v_QoS_Rules := valueof(p_QoS_Rules);

    } else if (p_GSM_MobilityInfo.IMS) {

      v_QoS_Rules := f_BuildDefaultQoSRules ("2"); // @sic R5-202585 sic@

    } // else use QoS 1 already defined above

    if (isvalue(p_QoSFlowDescr)) { // @sic R5-194896 sic@

      v_QoSFlowDescr := valueof(p_QoSFlowDescr);

    } else {

      if (p_GSM_MobilityInfo.IMS) {

        v_QoS_Flow := cs_QoS_Flow2(v_InterworkWithEPS, v_EPSBearerId); // @sic R5-204400 sic@ //Anritsu TTCN CR R5s201323

      } else { // Not IMS

        v_QoS_Flow := cs_QoS_Flow1(v_InterworkWithEPS, v_EPSBearerId); // @sic R5-204400 sic@ //Anritsu TTCN CR R5s201323

      }

        v_QoSFlowDescr := cs_QoSFlowDescr ({v_QoS_Flow});

    }

    if (isvalue(p_AlwaysOnPDUSessionInd)) {

      v_AlwaysOn := p_AlwaysOnPDUSessionInd;

    } else if (isvalue(p_GSM_MobilityInfo.AlwaysOn)) {

      v_AlwaysOn := cs_AlwaysOnNotAllowed;

    }

    v_PDN_TypeToBeUsed := valueof(v_PDU_SessionType.typeValue);

    select (v_PDN_TypeToBeUsed){

      case (tsc_PdnType_IPv4, tsc_PdnType_IPv6, tsc_PdnType_IPv4v6) {

        v_PDN_Address := f_PDN_AddressInfo_Get (f_NR5GC_GetPdnIndex(p_GSM_MobilityInfo)); // @sic R5s200643 sic@

        v_PDU_Address := f_NG_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(p_GSM_MobilityInfo.PCO), v_PDN_TypeToBeUsed, v_PDN_Address); //  @sic R5s200571, R5s200643 sic@

      }

    }

    if (isvalue(p_ExtdPCO)) {

      v_ExtdPCO := p_ExtdPCO;

    } else {

      v_ExtdPCO := f_NG_GetDefaultExtdProtocolConfigOptions (p_GSM_MobilityInfo.PCO, v_PDN_Address); // @sic R5s200643 sic@

    }

    v_GSMMsg := cs_NG_PDU_SESSION_ESTABLISHMENT_ACCEPT (cr_NG_PDU_SessionId(-, p_GSM_MobilityInfo.SessionId), // @sic R5s00564 sic@

                                                        p_GSM_MobilityInfo.PTI,

                                                        p_SSC_Mode,

                                                        v_PDU_SessionType,

                                                        v_QoS_Rules,

                                                        p_Session_AMBR,

                                                        p_Cause,

                                                        v_PDU_Address,

                                                        p_RQTimer,

                                                        p_S_NSSAI,

                                                        v_AlwaysOn,

                                                        v_MappedEPSBearerContexts, // @sic R5-195328 sic@

                                                        p_EAP,

                                                        v_QoSFlowDescr,

                                                        v_ExtdPCO,

                                                        v_DNN);

    return f_Get_NG_DLNASTransport (cs_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@

                                    v_GSMMsg,

                                    -,

                                    p_GSM_MobilityInfo.SessionId,  // @sic R5s190338, R5s190626, R5s200564 sic@

                                    omit,

                                    omit,

                                    p_BackOff);

  }
After change:

       import from TestcaseProperties all; //f_GetTestcaseAttrib_AdditionalPCSCF
  function f_Get_NG_PDUSessionEstablishmentAccept(GSM_MobilityInfo_Type p_GSM_MobilityInfo,

                                                  GMM_MobilityInfo_Type p_GMM_MobilityInfo,

                                                  NAS_PlmnId p_PLMN,

                                                  template (value) SSC_Mode p_SSC_Mode := cs_SSC_Mode(-, '001'B),

                                                  template (value) Session_AMBR p_Session_AMBR := cs_Session_AMBR ('06'O, '05'O, '0004'O, '05'O, '0004'O),

                                                  template (omit) PDU_SessionType p_PDU_SessionType := omit,

                                                  template (omit) GMM_GSM_Cause p_Cause := omit,

                                                  template (omit) GPRS_Timer p_RQTimer := omit,

                                                  template (omit) S_NSSAI_Type p_S_NSSAI := cs_S_NSSAI_SST1eMBB_WithIEI, // @sic R5s190919 sic@

                                                  template (omit) AlwaysOnPDUSessionInd p_AlwaysOnPDUSessionInd := omit,

                                                  template (omit) EAP_Message p_EAP := omit,

                                                  template (omit) QoS_Rules p_QoS_Rules := omit,

                                                  template (omit) QoSFlowDescr p_QoSFlowDescr := omit,

                                                  template (omit) ExtdProtocolConfigOptions p_ExtdPCO := omit,

                                                  template (omit) DNN p_DNN := omit,

                                                  template (omit) GPRS_Timer3 p_BackOff := omit) return template (value) NG_NAS_DL_Pdu_Type

  {

    var template (value) NG_NAS_DL_Message_Type v_GSMMsg;

    var template (value) PDU_SessionType v_PDU_SessionType;

    var B3_Type v_PDN_TypeToBeUsed;

    var PDN_AddressInfo_Type v_PDN_Address;

    var template (omit) PDU_Address v_PDU_Address := omit;

    var template (omit) AlwaysOnPDUSessionInd v_AlwaysOn := omit;

    var template (value) QoS_Rules v_QoS_Rules := f_BuildDefaultQoSRules ("1"); // @sic R5-202585 sic@

    var template (value) QoSFlowDescr v_QoSFlowDescr;

    var template (value) DNN v_DNN := cs_DNN(-, px_AccessPointName);

    var octetstring v_DNNValue;

    var template (omit) ExtdProtocolConfigOptions v_ExtdPCO;

    var template (omit) MappedEPSBearerContexts v_MappedEPSBearerContexts := omit;

    var template (value) MappedEPSContext v_MappedEPSContext;

    var bitstring v_EPSParam; // If used

    var O1_Type v_EPSBearerId; // If used

    var boolean v_InterworkWithEPS := false;

    var template (value) QoS_Flow v_QoS_Flow;

    if (isvalue(p_PDU_SessionType)) {

      v_PDU_SessionType := valueof(p_PDU_SessionType);

    } else {

      v_PDU_SessionType := cs_PDU_SessionType (p_GSM_MobilityInfo.SessionType);

    }

    if (isvalue(p_DNN)) { // test case is overwriting the DNN value

      v_DNN := valueof(p_DNN);

    } else if (pc_IMS_DNN_default or isvalue(p_GSM_MobilityInfo.DNN)) {   // if included with the Req

                                                      // @sic R5-204400 sic@ or not included but PICS = Default is IMS

      if (pc_IMS_DNN_default or p_GSM_MobilityInfo.IMS) { // overwrite value with complete response

        v_DNNValue := f_NG_GetDNNForIMS(p_GSM_MobilityInfo.DNN.dnnValue, p_PLMN);      //Anritsu TTCN CR R5s201327

      } else { // Not IMS, but included

        v_DNNValue := p_GSM_MobilityInfo.DNN.dnnValue;

      }

      v_DNN := cs_DNN(-, v_DNNValue);

    } // else use Default DNN already defined above if not included and not IMS

    if (isvalue(p_GMM_MobilityInfo.GMMCap)) { // @sic R5s190464 R5s191172 sic@

      if (p_GMM_MobilityInfo.GMMCap.s1Cap == '1'B) { // @sic R5-195328 sic@

        v_InterworkWithEPS := true;

        if (isvalue(p_GSM_MobilityInfo.EPS_Bearer)) { // @sic R5-204400 sic@

          v_EPSBearerId := hex2oct(int2hex(valueof(p_GSM_MobilityInfo.EPS_Bearer), 1) & '0'H);

        } else {

          FatalError(__FILE__, __LINE__, "EPS Supported but no bearer id calculated");

        }

        if (p_GSM_MobilityInfo.IMS) {

          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_2));

          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter(bit2oct(v_EPSParam)));

        } else {

          v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_1));

          v_MappedEPSContext := cs_OneMappedEPSContext(v_EPSBearerId, cs_EPSParameter(bit2oct(v_EPSParam)));

        }

        v_MappedEPSBearerContexts := cs_OneMappedEPSBearerContextList (v_MappedEPSContext);

      }

    }

    if (isvalue(p_QoS_Rules)) {

      v_QoS_Rules := valueof(p_QoS_Rules);

    } else if (p_GSM_MobilityInfo.IMS) {

      v_QoS_Rules := f_BuildDefaultQoSRules ("2"); // @sic R5-202585 sic@

    } // else use QoS 1 already defined above

    if (isvalue(p_QoSFlowDescr)) { // @sic R5-194896 sic@

      v_QoSFlowDescr := valueof(p_QoSFlowDescr);

    } else {

      if (p_GSM_MobilityInfo.IMS) {

        v_QoS_Flow := cs_QoS_Flow2(v_InterworkWithEPS, v_EPSBearerId); // @sic R5-204400 sic@ //Anritsu TTCN CR R5s201323

      } else { // Not IMS

        v_QoS_Flow := cs_QoS_Flow1(v_InterworkWithEPS, v_EPSBearerId); // @sic R5-204400 sic@ //Anritsu TTCN CR R5s201323

      }

        v_QoSFlowDescr := cs_QoSFlowDescr ({v_QoS_Flow});

    }

    if (isvalue(p_AlwaysOnPDUSessionInd)) {

      v_AlwaysOn := p_AlwaysOnPDUSessionInd;

    } else if (isvalue(p_GSM_MobilityInfo.AlwaysOn)) {

      v_AlwaysOn := cs_AlwaysOnNotAllowed;

    }

    v_PDN_TypeToBeUsed := valueof(v_PDU_SessionType.typeValue);

    select (v_PDN_TypeToBeUsed){

      case (tsc_PdnType_IPv4, tsc_PdnType_IPv6, tsc_PdnType_IPv4v6) {

        v_PDN_Address := f_PDN_AddressInfo_Get (f_NR5GC_GetPdnIndex(p_GSM_MobilityInfo)); // @sic R5s200643 sic@

        v_PDU_Address := f_NG_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(p_GSM_MobilityInfo.PCO), v_PDN_TypeToBeUsed, v_PDN_Address); //  @sic R5s200571, R5s200643 sic@

      }

    }

    if (isvalue(p_ExtdPCO)) {

      v_ExtdPCO := p_ExtdPCO;

    } else {

      if (f_GetTestcaseAttrib_AdditionalPCSCF (testcasename())) {

          v_ExtdPCO := f_NG_GetDefaultExtdProtocolConfigOptions (p_GSM_MobilityInfo.PCO, v_PDN_Address, true);

          } else {

          v_ExtdPCO := f_NG_GetDefaultExtdProtocolConfigOptions (p_GSM_MobilityInfo.PCO, v_PDN_Address, false);

          }
    }

    v_GSMMsg := cs_NG_PDU_SESSION_ESTABLISHMENT_ACCEPT (cr_NG_PDU_SessionId(-, p_GSM_MobilityInfo.SessionId), // @sic R5s00564 sic@

                                                        p_GSM_MobilityInfo.PTI,

                                                        p_SSC_Mode,

                                                        v_PDU_SessionType,

                                                        v_QoS_Rules,

                                                        p_Session_AMBR,

                                                        p_Cause,

                                                        v_PDU_Address,

                                                        p_RQTimer,

                                                        p_S_NSSAI,

                                                        v_AlwaysOn,

                                                        v_MappedEPSBearerContexts, // @sic R5-195328 sic@

                                                        p_EAP,

                                                        v_QoSFlowDescr,

                                                        v_ExtdPCO,

                                                        v_DNN);

    return f_Get_NG_DLNASTransport (cs_PayloadContainerType(tsc_PayloadContainerESMMsg), // @sic R5w190208 sic@

                                    v_GSMMsg,

                                    -,

                                    p_GSM_MobilityInfo.SessionId,  // @sic R5s190338, R5s190626, R5s200564 sic@

                                    omit,

                                    omit,

                                    p_BackOff);

  }
2.7 Change 7
	Function name
	f_NG_GetDefaultExtdProtocolConfigOptions

	Reason for change
	To accommodate change 6, modification of f_NG_GetDefaultExtdProtocolConfigOptions function needs to be done such that 2 PCO containers signalling 2 P-CSCF addresses.

	Summary of change
	In case of additional P=CSCF testcase there will be second PCO container added with P-CSCF server address as defined in PDN_3

	TTCN module
	Common\NG_NAS\NG_NASTemplateFunctions.ttcn

	MCC160 Comment
	


Before change:
     function f_NG_GetDefaultExtdProtocolConfigOptions(template (omit) ExtdProtocolConfigOptions p_Pco,

                                             PDN_AddressInfo_Type p_PDN_AddressInfo) return template (omit) ExtdProtocolConfigOptions

  {

    var template (omit) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList := omit;

    var ExtdProtocolConfigOptions v_ProtocolConfigOptionsRX;

    if (isvalue(p_Pco)) {

      v_ProtocolConfigOptionsRX := valueof(p_Pco);

      if (ispresent(p_Pco.pco)) {

        v_ProtocolContainerList := f_NG_NAS_GetExtdProtocolConfigOptionList(v_ProtocolConfigOptionsRX.pco, p_PDN_AddressInfo); // @sic R5s200643 sic@

      }

    }

    return f_NG_NAS_ExtdProtocolConfigOptionsTX(v_ProtocolContainerList);

  }
After change:

      function f_NG_GetDefaultExtdProtocolConfigOptions(template (omit) ExtdProtocolConfigOptions p_Pco,

                                             PDN_AddressInfo_Type p_PDN_AddressInfo,

                                             boolean p_AddadditionalPCSCF := false) return template (omit) ExtdProtocolConfigOptions

  {

    var template (omit) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList, v_ProtocolContainerList2 := omit;

    var ExtdProtocolConfigOptions v_ProtocolConfigOptionsRX;

    if (isvalue(p_Pco)) {

      v_ProtocolConfigOptionsRX := valueof(p_Pco);

      if (ispresent(p_Pco.pco)) {

        v_ProtocolContainerList := f_NG_NAS_GetExtdProtocolConfigOptionList(v_ProtocolConfigOptionsRX.pco, p_PDN_AddressInfo); // @sic R5s200643 sic@

          if (p_AddadditionalPCSCF) {

          v_ProtocolContainerList2 := f_NG_NAS_GetExtdProtocolConfigOptionList(v_ProtocolConfigOptionsRX.pco, f_PDN_AddressInfo_Get(PDN_3)); 

          v_ProtocolContainerList := v_ProtocolContainerList & v_ProtocolContainerList2;

          }
      }

    }

    return f_NG_NAS_ExtdProtocolConfigOptionsTX(v_ProtocolContainerList);

  }
2.8 Change 8

	Altstep name
	a_ApplicationCtrl

	Reason for change
	In case of additional P-CSCF testcase also the IMS server needs to be started for PDN_3 index in order to handle configuration messages.

	Summary of change
	In case of additional P-CSCF testcase the IMS server is started for PDN_3 index.

	TTCN module
	Common\IP_PTC\IP_PTC_Main.ttcn

	MCC160 Comment
	


Before change:
     altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;

    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartPDN)) {

        v_PdnIndex := v_IP_ConfigReq.StartPDN.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartPDN.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        select (v_PdnIndex) {

          case (PDN_1, PDN_2, PDN_2a) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (PDN_Internet, nrPDN_Internet, PDN_3) { // @sic R5s191074 sic@

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

          }

          case (ePDG_IMS1, ePDG_IMS2) {

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (ePDG_XCAP) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo, dnsServerOnly);

          }

          case (ePDG_Internet) {

            // nothing to be done

          }

        }
…
After change:

      altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;

    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartPDN)) {

        v_PdnIndex := v_IP_ConfigReq.StartPDN.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartPDN.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        select (v_PdnIndex) {

          case (PDN_1, PDN_2, PDN_2a) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (PDN_Internet, nrPDN_Internet, PDN_3) { // @sic R5s191074 sic@

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

            if (f_GetTestcaseAttrib_AdditionalPCSCF (testcasename())) {

               fl_IP_ImsServer_Start(v_PdnIndex);

            }
          }

          case (ePDG_IMS1, ePDG_IMS2) {

            fl_IP_ImsServer_Start(v_PdnIndex);

          }

          case (ePDG_XCAP) {

            fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo, dnsServerOnly);

          }

          case (ePDG_Internet) {

            // nothing to be done

          }

        }
…
2.9 Change 9

	Altstep name
	a_IP_IMS_Handler

	Reason for change
	In order to handle messages from IP component for the IMS PDN_3 the respective handlers representing IMS PDN_3 need to be started.

	Summary of change
	IMS PDN_3 handlers were added in IP to IMS handler

	TTCN module
	Common\IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change:
   altstep a_IP_IMS_Handler(inout IP_PTC_IMS_Info_Type p_IMS_Info) runs on IP_PTC

  {

    var IMS_Server_Type v_ImsServer_PDN1 := p_IMS_Info.Server[tsc_Index_IMS1];

    var IMS_Server_Type v_ImsServer_PDN2 := p_IMS_Info.Server[tsc_Index_IMS2];

    [] a_IP_IMS_UL_Handler(v_ImsServer_PDN1, tsc_Index_IMS1);

    [] a_IP_IMS_UL_Handler(v_ImsServer_PDN2, tsc_Index_IMS2);

    [] a_IP_IMS_DL_Handler(v_ImsServer_PDN1, tsc_Index_IMS1) {   // @sic R5s130454 sic@

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;

    }

    [] a_IP_IMS_DL_Handler(v_ImsServer_PDN2, tsc_Index_IMS2) {   // @sic R5s130454 sic@

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;

    }

    [] a_IP_IMS_Config(v_ImsServer_PDN1, IMS_CTRL[tsc_Index_IMS1]) {

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;     // configuration for PDN1 has been modified

    }

    [] a_IP_IMS_Config(v_ImsServer_PDN2, IMS_CTRL[tsc_Index_IMS2]) {

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;     // configuration for PDN2 has been modified

    }

    [] a_IP_IMS_TCP_ConnectionControl(v_ImsServer_PDN1) {

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;     // @sic R5s130378 sic@

    }

    [] a_IP_IMS_TCP_ConnectionControl(v_ImsServer_PDN2) {

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;     // @sic R5s130378 sic@

    }

  }
After change:

       altstep a_IP_IMS_Handler(inout IP_PTC_IMS_Info_Type p_IMS_Info) runs on IP_PTC

  {

    var IMS_Server_Type v_ImsServer_PDN1 := p_IMS_Info.Server[tsc_Index_IMS1];

    var IMS_Server_Type v_ImsServer_PDN2 := p_IMS_Info.Server[tsc_Index_IMS2];

    [] a_IP_IMS_UL_Handler(v_ImsServer_PDN1, tsc_Index_IMS1);

    [] a_IP_IMS_UL_Handler(v_ImsServer_PDN2, tsc_Index_IMS2);

    [ispresent(p_IMS_Info.Server[tsc_Index_IMS3])] a_IP_IMS_UL_Handler(p_IMS_Info.Server[tsc_Index_IMS3], tsc_Index_IMS3);
    [] a_IP_IMS_DL_Handler(v_ImsServer_PDN1, tsc_Index_IMS1) {   // @sic R5s130454 sic@

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;

    }

    [] a_IP_IMS_DL_Handler(v_ImsServer_PDN2, tsc_Index_IMS2) {   // @sic R5s130454 sic@

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;

    }

    [] a_IP_IMS_DL_Handler(p_IMS_Info.Server[tsc_Index_IMS3], tsc_Index_IMS3) {   

      p_IMS_Info.Server[tsc_Index_IMS3] := p_IMS_Info.Server[tsc_Index_IMS3];

    }

    [] a_IP_IMS_Config(v_ImsServer_PDN1, IMS_CTRL[tsc_Index_IMS1]) {

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;     // configuration for PDN1 has been modified

    }

    [] a_IP_IMS_Config(v_ImsServer_PDN2, IMS_CTRL[tsc_Index_IMS2]) {

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;     // configuration for PDN2 has been modified

    }

    [] a_IP_IMS_Config(p_IMS_Info.Server[tsc_Index_IMS3], IMS_CTRL[tsc_Index_IMS3]) {

      p_IMS_Info.Server[tsc_Index_IMS3] := p_IMS_Info.Server[tsc_Index_IMS3];     

    }

    [] a_IP_IMS_TCP_ConnectionControl(v_ImsServer_PDN1) {

      p_IMS_Info.Server[tsc_Index_IMS1] := v_ImsServer_PDN1;     // @sic R5s130378 sic@

    }

    [] a_IP_IMS_TCP_ConnectionControl(v_ImsServer_PDN2) {

      p_IMS_Info.Server[tsc_Index_IMS2] := v_ImsServer_PDN2;     // @sic R5s130378 sic@

    }

    [] a_IP_IMS_TCP_ConnectionControl(p_IMS_Info.Server[tsc_Index_IMS3]) {

      p_IMS_Info.Server[tsc_Index_IMS3] := p_IMS_Info.Server[tsc_Index_IMS3];     

    }

  }
2.10 Change 10
	Function name
	fl_SPI_GetValidValue

	Reason for change
	Currently there are no SPI values defined for all PDNs with index higher than 1 (e.g. for PDN_3), yet some are needed for the secure connection to be established. As a temporary solution we propose to assign the same SPI values as for PDNs with index=1  for any PDN indices higher than 1. This solution will work as long as two PDNs with the same SPIs will not be used by the same test (e.g. PDN_2 and PDN_3 would collide, but PDN_1 and PDN_3 would not)

	Summary of change
	Assigning the same SPI values for all PDN indeces higher than 1.

	TTCN module
	Common\IMS\IMS_Component.ttcn

	MCC160 Comment
	


Before change:
     function fl_SPI_GetValidValue(integer p_Offset) runs on IMS_PTC return integer

  {

    var integer v_Index := f_PDN_PdnIndex2ImsIndex(vc_IMS_Global.PdnIndex);    // @sic R5s201177: f_PDN_PdnIndex2ImsIndex instead of f_PDN_PdnIndex2Integer sic@

    var IntegerList_Type v_ValidValueList := vc_IMS_Global.ImsInfo.SPI_ValidValueList[v_Index];
    var integer v_ValidValueListLength := lengthof(v_ValidValueList);
    var integer v_SPI;

    if (v_ValidValueListLength > 0) {

      if (v_ValidValueListLength <= p_Offset) {

        FatalError(__FILE__, __LINE__, "Not enough SPI values");

      }

      v_SPI := v_ValidValueList[p_Offset];

    } else {

      v_SPI := 256 + (v_Index * 1024) + p_Offset; /* @sic R5s130651: v_Index used to get different SPIs for different PDNs sic@ */

    }

    if (v_SPI < 256) {  /* According to RFC 2406, SPI values from 0 to 255 are reserved */

      FatalError(__FILE__, __LINE__, "Invalid SPI value");

    }

    return v_SPI;

  }
After change:

      function fl_SPI_GetValidValue(integer p_Offset) runs on IMS_PTC return integer

  {

    var integer v_Index := f_PDN_PdnIndex2ImsIndex(vc_IMS_Global.PdnIndex);    // @sic R5s201177: f_PDN_PdnIndex2ImsIndex instead of f_PDN_PdnIndex2Integer sic@

    var IntegerList_Type v_ValidValueList;

    var integer v_ValidValueListLength;

    var integer v_SPI;

     if (v_Index > 1) {v_Index := 1};

    v_ValidValueList := vc_IMS_Global.ImsInfo.SPI_ValidValueList[v_Index];

    v_ValidValueListLength := lengthof(v_ValidValueList);
    if (v_ValidValueListLength > 0) {

      if (v_ValidValueListLength <= p_Offset) {

        FatalError(__FILE__, __LINE__, "Not enough SPI values");

      }

      v_SPI := v_ValidValueList[p_Offset];

    } else {

      v_SPI := 256 + (v_Index * 1024) + p_Offset; /* @sic R5s130651: v_Index used to get different SPIs for different PDNs sic@ */

    }

    if (v_SPI < 256) {  /* According to RFC 2406, SPI values from 0 to 255 are reserved */

      FatalError(__FILE__, __LINE__, "Invalid SPI value");

    }

    return v_SPI;

  }
2.11 Change 11

	Function name
	function f_NR5GC_IPCAN_MainLoop()

	Reason for change
	In case of additional P-CSCF testcase the main loop needs to be configured to handle IMS commands from IMS PDN3

	Summary of change
	In case of additional P-CSCF testcase the main loop has been configured to handle IMS commands from IMS PDN3

	TTCN module
	IMS\PART3_IPCAN\IPCAN_NR5GC.ttcn

	MCC160 Comment
	


Before change:
     function f_NR5GC_IPCAN_MainLoop() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_CellId := tsc_IMS_NR5GC_CellId;

    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;

    var IMS_IPCAN_Command_Type v_Command;

    var IMS_TestConfiguration_Type v_TestConfiguration;

    var TriggerResult_Type v_StartProcedureResult;

    var integer v_PdnIndex;

    var boolean v_ConnectionEstablished := false;

    var boolean v_Done := false;

    while (not v_Done) {

      alt {

        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_IMS1;

        }

        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_IMS2;

        }

      }
After change:

  import from TestcaseProperties all; //f_GetTestcaseAttrib_AdditionalPCSCF
      function f_NR5GC_IPCAN_MainLoop() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_CellId := tsc_IMS_NR5GC_CellId;

    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;

    var IMS_IPCAN_Command_Type v_Command;

    var IMS_TestConfiguration_Type v_TestConfiguration;

    var TriggerResult_Type v_StartProcedureResult;

    var integer v_PdnIndex;

    var boolean v_ConnectionEstablished := false;

    var boolean v_Done := false;

    while (not v_Done) {

      alt {

        [] IMS[tsc_Index_IMS1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_IMS1;

        }

        [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_IMS2;

        }

        [f_GetTestcaseAttrib_AdditionalPCSCF(testcasename())] IMS[tsc_Index_IMS3].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_IMS3;
        }

      }

…

2.12 Change 12

	Type name
	MTC_IMS_Components_Type

	Reason for change
	In order to accommodate change 1 in terms of creating additional IMS component representing the PDN_3 ther is a need to extend the MTC_IMS_Components_Type

	Summary of change
	MTC_IMS_Components_Type has been extended for IMS3

	TTCN module
	IMS\PART3_MTC\MTC_Component_IMS_Common.ttcn

	MCC160 Comment
	


Before change:
     type record MTC_IMS_Components_Type {      /* @status    APPROVED (IMS) */

    IMS_PTC  IMS1,

    IMS_PTC  IMS2,

    IP_PTC   IP

  };
After change:

      type record MTC_IMS_Components_Type {      /* @status    APPROVED (IMS) */

    IMS_PTC  IMS1,

    IMS_PTC  IMS2,

    IMS_PTC  IMS3,
    IP_PTC   IP

  };
2.13 Change 13
	Function name
	f_MTC_IMS_InitAndConnect_IP_IMS

	Reason for change
	In order to accommodate change 1 in terms of creation and correct port interconnection of IMS3 component number of changes needs to be done in f_MTC_IMS_InitAndConnect_IP_IMS

	Summary of change
	Extended f_MTC_IMS_InitAndConnect_IP_IMS function for IMS3 component, added its port interconnection with IMS1, IP and MTC components.

	TTCN module
	IMS\PART3_MTC\MTC_Component_IMS_Common.ttcn

	MCC160 Comment
	


Before change:
     function f_MTC_IMS_InitAndConnect_IP_IMS(SYSTEM_IMS_BASE p_System,

                                           IMS_PTC p_IMS1,

                                           IMS_PTC p_IMS2,

                                           boolean p_UseDNS := false,

                                           boolean p_UseXCAP := false) runs on MTC_IMS

  { /* @sic R5s141051 change 3: parameter p_UseXCAP sic@ */

    /* @sic R5s170448: p_UseDNS for WLAN sic@ */

    var IP_PTC v_IP;

    f_MTC_BASE_Map(p_System);

    v_IP := f_IP_PTC_CreateAndMap(p_System);

    vc_Components.IMS1 := p_IMS1;

    vc_Components.IMS2 := p_IMS2;

    vc_Components.IP   := v_IP;

    connect(p_IMS1:IMS_CTRL,     v_IP:IMS_CTRL[tsc_Index_IMS1]);

    connect(p_IMS1:IMS_Server,   v_IP:IMS_Server[tsc_Index_IMS1]);

    connect(p_IMS1:IMS_Client,   v_IP:IMS_Client[tsc_Index_IMS1]);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1], p_IMS1:MMI);

    if (p_IMS2 != null) {

      connect(p_IMS2:IMS_CTRL,   v_IP:IMS_CTRL[tsc_Index_IMS2]);

      connect(p_IMS2:IMS_Server, v_IP:IMS_Server[tsc_Index_IMS2]);

      connect(p_IMS2:IMS_Client, v_IP:IMS_Client[tsc_Index_IMS2]);

      connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2], p_IMS2:MMI);

      connect(p_IMS2:OtherPDN,   p_IMS1:OtherPDN);                /* @sic R5s130508 change 1 sic@ */

    }

    v_IP.start(f_IP_PTC_MainLoop(p_UseDNS, p_UseXCAP));           /* @sic R5-153795: support of ProSe - order of parameters changed sic@ */

  }
After change:

       function f_MTC_IMS_InitAndConnect_IP_IMS(SYSTEM_IMS_BASE p_System,

                                           IMS_PTC p_IMS1,

                                           IMS_PTC p_IMS2,

                                           boolean p_UseDNS := false,

                                           boolean p_UseXCAP := false,

                                           IMS_PTC p_IMS3 := null) runs on MTC_IMS

  { /* @sic R5s141051 change 3: parameter p_UseXCAP sic@ */

    /* @sic R5s170448: p_UseDNS for WLAN sic@ */

    var IP_PTC v_IP;

    f_MTC_BASE_Map(p_System);

    v_IP := f_IP_PTC_CreateAndMap(p_System);

    vc_Components.IMS1 := p_IMS1;

    vc_Components.IMS2 := p_IMS2;

    vc_Components.IMS3 := p_IMS3;

    vc_Components.IP   := v_IP;

    connect(p_IMS1:IMS_CTRL,     v_IP:IMS_CTRL[tsc_Index_IMS1]);

    connect(p_IMS1:IMS_Server,   v_IP:IMS_Server[tsc_Index_IMS1]);

    connect(p_IMS1:IMS_Client,   v_IP:IMS_Client[tsc_Index_IMS1]);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1], p_IMS1:MMI);

    if (p_IMS2 != null) {

      connect(p_IMS2:IMS_CTRL,   v_IP:IMS_CTRL[tsc_Index_IMS2]);

      connect(p_IMS2:IMS_Server, v_IP:IMS_Server[tsc_Index_IMS2]);

      connect(p_IMS2:IMS_Client, v_IP:IMS_Client[tsc_Index_IMS2]);

      connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2], p_IMS2:MMI);

      connect(p_IMS2:OtherPDN,   p_IMS1:OtherPDN);                /* @sic R5s130508 change 1 sic@ */

    }

    if (p_IMS3 != null) {

      connect(p_IMS3:IMS_CTRL,   v_IP:IMS_CTRL[tsc_Index_IMS3]);

      connect(p_IMS3:IMS_Server, v_IP:IMS_Server[tsc_Index_IMS3]);

      connect(p_IMS3:IMS_Client, v_IP:IMS_Client[tsc_Index_IMS3]);

      connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2], p_IMS3:MMI);

      connect(p_IMS3:OtherPDN,   p_IMS1:OtherPDN);                

    }
    v_IP.start(f_IP_PTC_MainLoop(p_UseDNS, p_UseXCAP));           /* @sic R5-153795: support of ProSe - order of parameters changed sic@ */

  }
2.14 Change 14
	Function name
	f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC

	Reason for change
	In order to accommodate change 1 in terms of creation and correct port interconnection of IMS3, the  component number of changes needs to be done in f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC.

	Summary of change
	Extended f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC function for IMS3 component, added its port interconnection with NR5GC.

	TTCN module
	IMS\PART3_MTC\MTC_Component_IMS_NR5GC.ttcn

	MCC160 Comment
	


Before change:
     function f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(SYSTEM_IMS p_System,

                                                   IMS_PTC p_IMS1,

                                                   IMS_PTC p_IMS2,

                                                   boolean p_UseDNS := false,

                                                   boolean p_UseXCAP := false) runs on MTC_IMS

  {

    var NR5GC_PTC v_NR5GC;

    var NASEMU_NRNG_PTC v_NASEMU_PTC;

    f_MTC_IMS_InitAndConnect_IP_IMS(p_System, p_IMS1, p_IMS2, p_UseDNS, p_UseXCAP);

    v_NR5GC := NR5GC_PTC.create;                                            // @sic R5s201263 sic@

    f_NR_BASE_PTC_Map(p_System, v_NR5GC);                                   // map NR5GC system ports

    v_NASEMU_PTC := f_NasEmu5G_CreateConnectAndMap_NRNG(p_System, v_NR5GC); // create NasEmu, map system ports and connect with NR5GC PTC

    v_NASEMU_PTC.start(f_NASEMU_MainLoop_NRNG());                           // start NASEMU

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_NR], v_NR5GC:UT);

    connect(v_NR5GC:IP, vc_Components.IP:NR_CTRL);

    if (p_IMS1 != null) { connect(vc_Components.IMS1:IPCAN, v_NR5GC:IMS[tsc_Index_IMS1]); }

    if (p_IMS2 != null) { connect(vc_Components.IMS2:IPCAN, v_NR5GC:IMS[tsc_Index_IMS2]); }

    f_UT_InitialPowerOffUE(Ut);

    v_NR5GC.start(f_NR5GC_IPCAN_MainLoop());

  }
After change:

      function f_MTC_IMS_CreateMapAndConnectPTCs_NR5GC(SYSTEM_IMS p_System,

                                                   IMS_PTC p_IMS1,

                                                   IMS_PTC p_IMS2,

                                                   boolean p_UseDNS := false,

                                                   boolean p_UseXCAP := false,

                                                   IMS_PTC p_IMS3 := null) runs on MTC_IMS

  { 

    var NR5GC_PTC v_NR5GC;

    var NASEMU_NRNG_PTC v_NASEMU_PTC;

    f_MTC_IMS_InitAndConnect_IP_IMS(p_System, p_IMS1, p_IMS2, p_UseDNS, p_UseXCAP, p_IMS3);

    v_NR5GC := NR5GC_PTC.create;                                            // @sic R5s201263 sic@

    f_NR_BASE_PTC_Map(p_System, v_NR5GC);                                   // map NR5GC system ports

    v_NASEMU_PTC := f_NasEmu5G_CreateConnectAndMap_NRNG(p_System, v_NR5GC); // create NasEmu, map system ports and connect with NR5GC PTC

    v_NASEMU_PTC.start(f_NASEMU_MainLoop_NRNG());                           // start NASEMU

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_NR], v_NR5GC:UT); 

    connect(v_NR5GC:IP, vc_Components.IP:NR_CTRL);

    if (p_IMS1 != null) { connect(vc_Components.IMS1:IPCAN, v_NR5GC:IMS[tsc_Index_IMS1]); }

    if (p_IMS2 != null) { connect(vc_Components.IMS2:IPCAN, v_NR5GC:IMS[tsc_Index_IMS2]); }

    if (p_IMS3 != null) { connect(vc_Components.IMS3:IPCAN, v_NR5GC:IMS[tsc_Index_IMS3]); }
    f_UT_InitialPowerOffUE(Ut);

    v_NR5GC.start(f_NR5GC_IPCAN_MainLoop());

  }
2.15 Change 15
	Function name
	f_TC_6_2_NR5GC_IMS3

	Reason for change
	1) Based on the reason for change 1 and the associated prose CR, the steps 4-15 happen on the additional IMS3 component associated with PDN3. Therefore a new f_TC_6_2_NR5GC_IMS3 function is added in IMS_NR5GC_RegistrationTestcases.ttcn to represent IMS handling on PDN3 which takes over the test body handling of steps 4-15 from f_TC_6_2_NR5GC_IMS1. Additionally, some coordination between IMS1 and IMS3 components need to be added.
2) Based on the reason for change 1 and the associated prose CR  the REGISTER messages at step 4, 6 and 8 will be sent over a new dialog and therefore the CSEQ check should be removed. Only single CSEQ check shall be kept – the one between step 8 and 10.
3) After step 5 it is critical to check that REGISTER at step 6 is sent at least 10s later. However in case the UE continues with step 6 in exactly 10s then it may collide with delays caused by functions f_IMS_CC_ReleaseConnection and f_IMS_CC_StartSignalling, therefore we propose to remove RRC release between steps 5 and 6 as it has no effect on the test purpose anyway and may otherwise cause unfair test failure.
#R5-205581 raised at RAN5meeting#89-e is associated with this CR. 

	Summary of change
	1) Steps 4-15 are now handled by f_TC_6_2_NR5GC_IMS3 instead of by f_TC_6_2_NR5GC_IMS1. Coordination between IMS1 and IMS3 has been added in order to hand over control to IMS3 for steps 4-15, to signal IMS1 that the test ends and to transfer routing parameters between IMS1 and IMS3 component.
2) CSEQ check has been removed for all steps except for step 10.

3) Removed RRC connection release between steps 5 and 6 and replaced the 10s delay with an “alt” checking for an early reception of step 6 REGISTER.

	TTCN module
	IMS\NR5GC_6\IMS_NR5GC_RegistrationTestcases.ttcn

	MCC160 Comment
	


Before change:
     function f_TC_6_2_NR5GC_IMS1() runs on IMS_PTC

  {

    var IMS_DATA_REQ v_IMS_REGISTER_REQ;

    var REGISTER_Request v_RegisterReq;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfoDL;

    var template (omit) RetryAfter v_RetryAfterForREGISTER;

    var integer v_RetryAfterForPublish := 60;    // as per "503 Service Unavailable" default message contents (34.229-1 A.4.2)

    var charstring v_RegisterExpiration;

    var integer v_Cseq;

    timer t_WaitForRegister;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security);

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    // @siclog "Step 2" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest();

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    v_RoutingInfoDL := f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ.RoutingInfo);

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq);

    // @siclog "Step 3" siclog@

    v_RetryAfterForREGISTER := omit;

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine503, f_IMS_CommonResponse_503_MessageHeaderTX(v_RegisterReq.msgHeader, v_RetryAfterForREGISTER))));

    f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);

    // @siclog "Step 4" siclog@

    t_WaitForRegister.start(f_IMS_SetTimerToleranceMax(300.0));         // 5min

    alt {

      [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request())) -> value v_IMS_REGISTER_REQ {

        t_WaitForRegister.stop;

        v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

        v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);

        f_IMS_REGISTER_InitialRequest(-, v_IMS_REGISTER_REQ);           // common checks

        f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 4");

      }

      [] t_WaitForRegister.timeout {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: timeout");

      }

    }

    // @siclog "Step 5" siclog@

    v_RetryAfterForREGISTER := cs_RetryAfter(10);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine503, f_IMS_CommonResponse_503_MessageHeaderTX(v_RegisterReq.msgHeader, v_RetryAfterForREGISTER))));

    f_IMS_CC_ReleaseConnection(IPCAN_MO_IMS_Signalling);

    f_IMS_CC_StartSignalling(IPCAN_MO_IMS_Signalling);
    // @siclog "Step 6" siclog@

    f_Delay(f_IMS_SetTimerToleranceMin(10.0));

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest();

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;
    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // @siclog "Step 7" siclog@

    v_RegisterExpiration := "800000";

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine423, f_IMS_RegisterResponse_423_MessageHeaderTX(v_RegisterReq, v_RegisterExpiration))));

    // @siclog "Step 8" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest(-, -, -, v_RegisterExpiration);

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9" siclog@

    f_IMS_Register_SecurityInit(v_RegisterReq);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 10" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, -, -, v_RegisterExpiration);

    v_RoutingInfoDL := f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);

    // @siclog "Step 11" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq, -, v_RegisterExpiration))));

    // @siclog "Step 12 - 15" siclog@

    f_IMS_Register_SubscribeNotify(v_IMS_REGISTER_REQ, tsc_IMS_PublishDuringSubscribeForRegEvent, v_RetryAfterForPublish);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

  }
After change:

      function f_TC_6_2_NR5GC_IMS1() runs on IMS_PTC //This is to handle PDN_1

  {

    var IMS_DATA_REQ v_IMS_REGISTER_REQ;

    var REGISTER_Request v_RegisterReq;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfoDL;

    var template (omit) RetryAfter v_RetryAfterForREGISTER;

    var integer v_Cseq;

    f_IMS_CC_Preamble(IPCAN_SignallingOnly, IMS_NULL, IMS_Security); //PDN_1 - default 6th parameter

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartSignalling(IPCAN_InitialRegistration);

    // @siclog "Step 2" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest();

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    v_RoutingInfoDL := f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ.RoutingInfo);

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq);

    // @siclog "Step 3" siclog@

    v_RetryAfterForREGISTER := omit;

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine503, f_IMS_CommonResponse_503_MessageHeaderTX(v_RegisterReq.msgHeader, v_RetryAfterForREGISTER))));

    //Step 4-15 goes on PDN3

    f_IMS_IMS_SendCoOrdMsg(OtherPDN, cms_IMS_IMS_Data(v_RoutingInfoDL, omit, omit));

    //Step 6-15 to happen on PDN1

    f_IMS_IMS_WaitForTrigger(OtherPDN);
    f_IMS_TestBody_Set(false);

  }

  function f_TC_6_2_NR5GC_IMS3() runs on IMS_PTC

  {

    var IMS_DATA_REQ v_IMS_REGISTER_REQ;

    var REGISTER_Request v_RegisterReq;

    var template (omit) RetryAfter v_RetryAfterForREGISTER := omit;

    var charstring v_RegisterExpiration;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfoDL;

    var IMS_IMS_Data_Type v_DataRx;

    var integer v_Cseq;

    var integer v_RetryAfterForPublish := 60;    // as per "503 Service Unavailable" default message contents (34.229-1 A.4.2)

    timer t_WaitForRegister;

    //Initialise PTC for PDN3

    f_IMS_PTC_Init(PDN_3, -, IMS_Security, omit, px_DigestPasswordForSIP);

    //Routing information to be taken from IMS1

    v_DataRx := f_IMS_IMS_WaitForTrigger(OtherPDN);

    v_RoutingInfoDL := v_DataRx.IMS_Routing;
    // @siclog "Step 4" siclog@

    t_WaitForRegister.start(f_IMS_SetTimerToleranceMax(300.0));         // 5min 

    alt {

      [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request())) -> value v_IMS_REGISTER_REQ {

        t_WaitForRegister.stop;

        v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

        //CSEQ shall not be checked
        //v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq); 

        f_IMS_REGISTER_InitialRequest(-, v_IMS_REGISTER_REQ);           // common checks

        f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 4");

      }

      [] t_WaitForRegister.timeout {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4: timeout");

      }

    }

    // @siclog "Step 5" siclog@

    v_RetryAfterForREGISTER := cs_RetryAfter(10);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine503, f_IMS_CommonResponse_503_MessageHeaderTX(v_RegisterReq.msgHeader, v_RetryAfterForREGISTER))));

    t_WaitForRegister.start(f_IMS_SetTimerToleranceMin(10.0));

    alt {

      [] IMS_Server.receive(car_IMS_Register_Request(cr_REGISTER_Request())) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6: REGISTER received too early");

      }

      [] t_WaitForRegister.timeout  {

         // @siclog "Step 6" siclog@

         v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest();

         f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 6");

      }

    }
    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;
    //Cseq not checked
    //v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // @siclog "Step 7" siclog@

    v_RegisterExpiration := "800000";

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine423, f_IMS_RegisterResponse_423_MessageHeaderTX(v_RegisterReq, v_RegisterExpiration))));

    // @siclog "Step 8" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_InitialRequest(-, -, -, v_RegisterExpiration);

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    //CSEQ shall be checked only between steps 8 and 10

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 8");

    // @siclog "Step 9" siclog@

    f_IMS_Register_SecurityInit(v_RegisterReq);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    // @siclog "Step 10" siclog@

    v_IMS_REGISTER_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, -, -, v_RegisterExpiration);

    v_RoutingInfoDL := f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ.RoutingInfo);

    v_RegisterReq := v_IMS_REGISTER_REQ.Request.Register;

    v_Cseq := fl_TC_6_2_CheckGetCseq(v_RegisterReq, v_Cseq);

    // @siclog "Step 11" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfoDL, cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq, -, v_RegisterExpiration))));

    // @siclog "Step 12 - 15" siclog@

    f_IMS_Register_SubscribeNotify(v_IMS_REGISTER_REQ, tsc_IMS_PublishDuringSubscribeForRegEvent, v_RetryAfterForPublish);

    f_IMS_CC_Postamble(IPCAN_MO_IMS_Signalling);

    //Signal to IMS1 that this is the end of test

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }
2.16 Change 16
	Function name
	f_IMS_CC_Preamble

	Reason for change
	Extension of f_IMS_CC_Preamble function by the PDN index parameter will allow to initialise IMS components for other than just PDN_1. This change is needed for change 15.

	Summary of change
	Extension of f_IMS_CC_Preamble function by the PDN index parameter.

	TTCN module
	IMS\Common\IMS_CC_CommonFunctions.ttcn

	MCC160 Comment
	


Before change:
     function f_IMS_CC_Preamble(IMS_TestConfiguration_Type p_TestConfiguration,

                             IMS_InitState_Type p_InitState,

                             template (omit) IMS_SecurityScheme_Type p_SecurityScheme := omit,

                             template (omit) IPCAN_RAN_Type p_RanType := omit,

                             boolean p_SigCompIsActive := false) runs on IMS_PTC

  { /* p_TestProcedure  ..  major test procedure of the test case (=> number/type of DRBs to be configured at SS

     * p_InitState      ..  IDLE => intitial registration including IMS registration is done; RRC connection is released

     *                      NULL => intitial registration is started at the IPCAN but IMS registration shall be controled by the test case */

    /* @sic R5s130133 additional changes: XRES length shall be queried by the IMS PTC i.e. is not returned by f_IMS_IPCAN_StartProcedure_Preamble anymore sic@ */

    /* @sic R5s130347: local GuardTimer sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA replaced by p_SecurityScheme sic@ */

    /* @sic R5w160208 FBBA: new parameter p_RanType sic@ */

    timer t_GuardTimer := tsc_GuardTimePreamble;

    t_GuardTimer.start;

    f_IMS_PTC_Init(PDN_1, -, p_SecurityScheme, p_RanType, px_DigestPasswordForSIP);      /* local initialisation */

    f_IMS_IPCAN_Init(IPCAN, p_TestConfiguration);               /* initialisation of IPCAN:

                                                                   - RAN type is selected by IPCAN PTC acc. to the PIXITs

                                                                   - cell is created acc. to the RAN type and PIXITs

                                                                   - Test procedure is handed over as for e.g. EUTRA we need to know in advance which DRBs need to be (pre-)configured at the SS

                                                                   p_TestcaseName is present If the test case requires a specific IPCAN initialisation e.g. for cell barring */

    if (p_InitState == IMS_REGISTERED) {

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_InitialRegistration);

      f_IMS_Registration(-, -, p_SigCompIsActive);

      f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    }

    t_GuardTimer.stop;

  }
After change:

     function f_IMS_CC_Preamble(IMS_TestConfiguration_Type p_TestConfiguration,

                             IMS_InitState_Type p_InitState,

                             template (omit) IMS_SecurityScheme_Type p_SecurityScheme := omit,

                             template (omit) IPCAN_RAN_Type p_RanType := omit,

                             boolean p_SigCompIsActive := false,

                             PDN_Index_Type p_PdnIndex := PDN_1) runs on IMS_PTC

  { /* p_TestProcedure  ..  major test procedure of the test case (=> number/type of DRBs to be configured at SS

     * p_InitState      ..  IDLE => intitial registration including IMS registration is done; RRC connection is released

     *                      NULL => intitial registration is started at the IPCAN but IMS registration shall be controled by the test case */

    /* @sic R5s130133 additional changes: XRES length shall be queried by the IMS PTC i.e. is not returned by f_IMS_IPCAN_StartProcedure_Preamble anymore sic@ */

    /* @sic R5s130347: local GuardTimer sic@ */

    /* @sic R5s130756 additional changes: p_IsGIBA replaced by p_SecurityScheme sic@ */

    /* @sic R5w160208 FBBA: new parameter p_RanType sic@ */

    timer t_GuardTimer := tsc_GuardTimePreamble;

    t_GuardTimer.start;

    f_IMS_PTC_Init(p_PdnIndex, -, p_SecurityScheme, p_RanType, px_DigestPasswordForSIP);      /* local initialisation */

    f_IMS_IPCAN_Init(IPCAN, p_TestConfiguration);               /* initialisation of IPCAN:

                                                                   - RAN type is selected by IPCAN PTC acc. to the PIXITs

                                                                   - cell is created acc. to the RAN type and PIXITs

                                                                   - Test procedure is handed over as for e.g. EUTRA we need to know in advance which DRBs need to be (pre-)configured at the SS

                                                                   p_TestcaseName is present If the test case requires a specific IPCAN initialisation e.g. for cell barring */

    if (p_InitState == IMS_REGISTERED) {

      f_IMS_IPCAN_StartProcedure(IPCAN, IPCAN_InitialRegistration);

      f_IMS_Registration(-, -, p_SigCompIsActive);

      f_IMS_CC_ReleaseConnection(IPCAN_InitialRegistration);

    }

    t_GuardTimer.stop;

  }  
2.17 Change 17

	Type name
Template name
	IMS_IMS_Data_Type
cms_IMS_IMS_Data

	Reason for change
	There are certain parameters that need to be exchanged between IMS1 and IMS3 components. For this reason changes 17, 18 and 19 are proposed.

	Summary of change
	New type as well as new template for exchanging data between IMS components has been created.

	TTCN module
	IMS\IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	


After change:

import from IMS_ASP_TypeDefs all;
  type record IMS_IMS_Data_Type {

    IMS_RoutingInfo_Type       IMS_Routing                 optional,

    SecurityClientParams_Type    SecurityClientParams        optional,

    charstring                 CallId                      optional

  };
  template (value) IMS_IMS_Coordination_MSG cms_IMS_IMS_Data(IMS_RoutingInfo_Type    p_IMS_Routing,

                                                             SecurityClientParams_Type p_SecurityClientParams,

                                                             charstring              p_CallId) :=

  { IMS_Data := {

       IMS_Routing := p_IMS_Routing,

       SecurityClientParams := p_SecurityClientParams,

       CallId := p_CallId

       }

  };

  template IMS_IMS_Coordination_MSG cmr_IMS_IMS_Data :=

  { 

    IMS_Data := *

  };
2.18 Change 18

	Function name
	f_IMS_IMS_WaitForTrigger

	Reason for change
	In case some data are needed for exchange between IMS components, there is a need to extend f_IMS_IMS_WaitForTrigger to be able to receive them.

	Summary of change
	f_IMS_IMS_WaitForTrigger function has been extended to receive eventual data.

	TTCN module
	IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	


Before change:
     function f_IMS_IMS_WaitForTrigger(IMS_IMS_CO_ORD_PORT p_Port)

  {

    p_Port.receive(cmr_IMS_IMS_Trigger);
  }
After change:

  function f_IMS_IMS_WaitForTrigger(IMS_IMS_CO_ORD_PORT p_Port) return IMS_IMS_Data_Type

  {

    var IMS_IMS_Coordination_MSG v_ReceivedMsg;

    var IMS_IMS_Data_Type v_DataRx;

    //Receive either simple trigger or some IMS data

    alt {

      [] p_Port.receive(cmr_IMS_IMS_Data) -> value v_ReceivedMsg {

      v_DataRx := v_ReceivedMsg.IMS_Data;      

      }

      [] p_Port.receive(cmr_IMS_IMS_Trigger) {

      v_DataRx := omit;      

      }

    }

    return v_DataRx;
  }
2.19 Change 19

	Type name
	IMS_IMS_Coordination_MSG

	Reason for change
	Hand in hand with changes 17 and 18 the IMS_IMS_Coordination_MSG type needs to be extended.

	Summary of change
	IMS_IMS_Coordination_MSG type has been extended.

	TTCN module
	Common\IMS\IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	


Before change:
  type union IMS_IMS_Coordination_MSG 

    Null_Type                           TriggerEvent

  };
After change:

    type union IMS_IMS_Coordination_MSG {                 

    Null_Type                           TriggerEvent,

    IMS_IMS_Data_Type                   IMS_Data
  };
2.20 Change 20

	Type name
	SecurityClientParams_Type

	Reason for change
	The type SecurityClientParams_Type declaration has to be moved from IMS_Procedures_RegistrationCommon.ttcn to IMS_PTC_CoordMsg.ttcn in order to avoind recursive imports.

	Summary of change
	.

	TTCN module
	IMS_Procedures_RegistrationCommon.ttcn
IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	


Before change:
//Originally in IMS_Procedures_RegistrationCommon.ttcn
  type record SecurityClientParams_Type {   /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, NR5GC_IRAT, POS) */

    PortNumber_Type Port_us,

    PortNumber_Type Port_uc,

    IPsec_SPI_Type  SPI_us,

    IPsec_SPI_Type  SPI_uc

  };
After change:

       import from IP_ASP_TypeDefs all;
//Newly in IMS_PTC_CoordMsg.ttcn
  type record SecurityClientParams_Type {   /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, NR5GC_IRAT, POS) */

    PortNumber_Type Port_us,

    PortNumber_Type Port_uc,

    IPsec_SPI_Type  SPI_us,

    IPsec_SPI_Type  SPI_uc

  };
2.21 Change 21

	Function name
	

	Reason for change
	Due to change 20, the IMS_CC_Authentication.ttcn needs to import IMS_PTC_CoordMsg as it references to SecurityClientParams_Type

	Summary of change
	.Added import from IMS_PTC_CoordMsg

	TTCN module
	IMS\Common\IMS_CC_Authentication.ttcn

	MCC160 Comment
	


Before change:
After change:

       import from IMS_PTC_CoordMsg all;  
3 Execution Log Files 
3.1 Qualcomm SM8350
The Qualcomm SM8350 5G IMS UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]. In the log file (in .xml format) the complete test case execution can be seen using band n4.:

· Test case execution log file:

TC_6_2_LOG.xml
· PICS/PIXIT parameter file:
TC_6_2_PIXIT.txt

